MATH 357 -- Combinatorics
Some Generating Function Computations
March 17, 2017

These computations show how to determine

1. The number of integer solutions of the equations

$$x\_1 + x\_2 + x\_3 + 2\,x\_4 = 7$$
$$2\,x\_1 + x\_2 + 4\,x\_3 + 5\,x\_4 = 22$$

2. Then the number of integer solutions of the inequalities:

$$x\_1 + x\_2 + x\_3 + 2\,x\_4 \le 7$$
$$2\,x\_1 + x\_2 + 4\,x\_3 + 5\,x\_4 \le 22$$

We'll start with the equalities. Note first that the equalities imply that $x\_1 \le 7, x\_2 \le 7, x\_3 \le 5, x\_4 \le 3$ This means that we only need to consider the *truncated generating function:*

$$\left(1 + u\cdot v^2 + \ldots + \left(u\cdot v^2\right)^7\right)\cdot\left(1 + u\cdot v + \ldots + (u\cdot v)^7\right)\cdot\left(1 + u\cdot v^4 + \ldots + \left(u\cdot v^4\right)^5\right)\cdot\left(1 + u^2\cdot v^5 + \ldots + \left(u^2\cdot v^5\right)^3\right)$$

Here's one way to compute that. We start by creating a function that will expand one of the truncated geometric series:

$restart;$
$geom := k \to sum(x^i,\ i = 0..k);$

$$k \to \sum_{i=0}^{k} x^i \qquad \textbf{(1)}$$

$m12 := subs\left(x = u\cdot v^2,\ eval(geom(7))\right);$

$$u^7\,v^{14} + u^6\,v^{12} + u^5\,v^{10} + u^4\,v^8 + u^3\,v^6 + u^2\,v^4 + u\,v^2 + 1 \qquad \textbf{(2)}$$

$m11 := subs(x = u\cdot v,\ eval(geom(7)));$
$m14 := subs\left(x = u\cdot v^4,\ eval(geom(5))\right);$
$m25 := subs\left(x = u^2\cdot v^5,\ eval(geom(3))\right);$

$$u^7\,v^7 + u^6\,v^6 + u^5\,v^5 + u^4\,v^4 + u^3\,v^3 + u^2\,v^2 + u\,v + 1$$
$$u^5\,v^{20} + u^4\,v^{16} + u^3\,v^{12} + u^2\,v^8 + u\,v^4 + 1$$
$$u^6\,v^{15} + u^4\,v^{10} + u^2\,v^5 + 1 \qquad \textbf{(3)}$$

$gf := expand(m12\cdot m11\cdot m14\cdot m25):$

$coeff(coeff(gf, u^7), v^{22});$

$$4 \tag{4}$$

So the number of solutions of the equalities is 4. For the inequalities, we basically want to truncate this generating function again, removing all $u^m \cdot v^n$ where $m > 7$ or $n > 22$

```
tgf := gf:
for i to nops(gf) do
   if degree(op(i, gf), u) > 7 or degree(op(i, gf), v) > 22 then
     tgf := expand(tgf - op(i, gf));
   end if;
 end do:
nops(tgf);
```

$$79 \tag{5}$$

$tgf;$

$$4u^7v^{22} + 2u^7v^{21} + u^6v^{22} + 4u^7v^{20} + 2u^6v^{21} + 6u^7v^{19} + u^6v^{20} + 4u^7v^{18} + 2u^6v^{19} + u^5v^{20} + 6u^7v^{17} + 4u^6v^{18} + 8u^7v^{16} + 2u^6v^{17} + u^5v^{18} + 5u^7v^{15} + 4u^6v^{16} + 2u^5v^{17} + 6u^7v^{14} + 6u^6v^{15} + u^5v^{16} + 6u^7v^{13} + 4u^6v^{14} + 2u^5v^{15} + u^4v^{16} + 3u^7v^{12} + 5u^6v^{13} + 4u^5v^{14} + 3u^7v^{11} + 6u^6v^{12} + 2u^5v^{13} + u^4v^{14} + 3u^7v^{10} + 3u^6v^{11} + 4u^5v^{12} + 2u^4v^{13} + u^7v^9 + 3u^6v^{10} + 5u^5v^{11} + u^4v^{12} + u^7v^8 + 3u^6v^9 + 3u^5v^{10} + 2u^4v^{11} + u^3v^{12} + u^7v^7 + u^6v^8 + 3u^5v^9 + 4u^4v^{10} + u^6v^7 + 3u^5v^8 + 2u^4v^9 + u^3v^{10} + u^6v^6 + u^5v^7 + 3u^4v^8 + 2u^3v^9 + u^5v^6 + 3u^4v^7 + u^3v^8 + u^5v^5 + u^4v^6 + 2u^3v^7 + u^2v^8 + u^4v^5 + 3u^3v^6 + u^4v^4 + u^3v^5 + u^2v^6 + u^3v^4 + 2u^2v^5 + u^3v^3 + u^2v^4 + u^2v^3 + uv^4 + u^2v^2 + uv^2 + uv + 1 \tag{6}$$

$subs(u = 1, v = 1, tgf);$

$$192 \tag{7}$$